Gulf General Atomic MASTER
Incorporated
P.O. Box 608, San Diego, California 92112

GAMD-8718

A ONE-GROUP TWO-DIMENSIONAL CRITICAL SIZE PREDICTION SUBROUTINE

Work done by:

D. R. Mathews

Report written by:

D. R. Mathews

Prepared under
Contract AT(04-3)-167, P.A. #14 and #37
for the
U.S. Atomic Energy Commission

Projects:

278.31 and 4010

May 23, 1968

CONTENTS

ABSTRACT

A one-energy group diffusion theory subroutine has been developed to predict critical sizes of reflected finite cylindrical reactors for use in parametric surveys. An iteration between axial and radial one-dimensional reflected calculations is performed to estimate the critical size.

TABLES

SYMBOLS

α	core height-to-diameter ratio
ρ_o	core radius, cm
δ_a	axial reflector savings, cm
δ_r	radial reflector savings, cm
$\nu\Sigma_{f_c}$	nu times the macroscopic fission cross section in the core region, n/f-cm
Σ_{a_c}	macroscopic absorption cross section in the core region, cm^{-1}
D_c	diffusion coefficient in the core region, cm^{-1}
δ_p	plane geometry reflector savings, cm
D_r	average reflector diffusion coefficient, cm^{-1}
T	average reflector thickness, cm
Σ_{a_r}	average reflector macroscopic absorption cross section, cm^{-1}
D_{rr}	radial reflector diffusion coefficient, cm^{-1}
Σa_{rr}	radial reflector macroscopic absorption cross section, cm^{-1}
T_r	radial reflector thickness, cm
D_{ar}	axial reflector diffusion coefficient, cm^{-1}
Σa_{ar}	axial reflector macroscopic absorption cross section, cm^{-1}
T_a	axial reflector thickness, cm.

1. INTRODUCTION

A subroutine has been written to estimate the critical size of a cylindrical reflected reactor from a one-energy group diffusion theory model using an iterative process in which the side-reflected and end-reflected cylindrical geometry problems are solved alternately until converged values of the bucklings (reflector savings) are obtained for both the axial and radial directions.

The subroutine has been incorporated into a parametric survey code used in thermionic reactor design studies.(1) The subroutine is intended to provide data relative to a specific design point and may be seriously in error in absolute magnitude if used without normalization.

The core region is assumed to consist of a homogeneous right circular cylinder. The axial and radial reflectors may have different thicknesses and/or materials.

The subroutine makes use of nine additional non-standard subroutines. Six of these ten subroutines compute the Bessel functions that are required in the solution of the side-reflected cylindrical geometry problem, one calculates the axial reflector savings, one calculates the core radius for cores with height-to-diameter ratios greater than three, and one calculates the radial reflector savings. Thus the complete subroutine package consists of ten subprograms. A common block labeled "CØM1" is used to transmit information between the subprograms. The program that calls this group of subprograms must not have a common block with the same label.

All of the necessary input to the subroutine is transmitted explicitly via eleven arguments and the value of the estimated critical diameter is returned via the twelfth argument. The program listing in the Appendix contains the detailed specification of the input argument list. Note that all dimensions are in centimeters.

The Bessel function subprograms are based on the polynomial approximations given in the N.B.S. Handbook[2] and were found to give at least seven significant figure accuracy over a range of arguments from 0.1 to 17.0.

2. CRITICAL SIZE CALCULATION

For a cylindrical core of given radius, given radial reflector thickness, and given material properties, the requirement that the system be just critical reduces to a statement concerning the axial leakage required for criticality. Similarly, for a cylindrical core of given height, given axial reflector thickness, and given material properties, one may solve for the radial leakage for which the system will be just critical. This is true regardless of whether one uses one-group diffusion theory, multi-group diffusion theory or very refined transport theory.

For simplicity and speed in survey calculations we have chosen to use one-group diffusion theory at this point. The one-group criticality relation may be written in the form

$$B^2 = \left(\frac{1.5708}{\alpha\rho_o + \delta_a}\right)^2 + \left(\frac{2.405}{\rho_o + \delta_r}\right)^2 \qquad (2.1.)$$

where

$$B^2 \equiv \frac{\nu\Sigma_{f_c} - \Sigma_{a_c}}{D_c} \qquad (2.2.)$$

The calculation of the critical core radius ρ_o proceeds in three stages. First, the axial and radial reflector savings are set equal to the plane geometry reflector savings δ_p calculated from

$$\delta_p = \frac{1}{B}\left\{\tan^{-1}\left(\frac{D_c B}{D_r k} \tanh (kT)\right)\right\} \qquad (2.3.)$$

where

$$k^2 \equiv \Sigma_{a_r}/D_r \qquad (2.4.)$$

At this stage, the average reflector thickness T, diffusion coefficient D_r, and macroscopic absorption cross section Σa_r are used. The average reflector thickness also includes an extrapolation length taken to be 2.13 D_r. An initial estimate of the critical radius is now obtained by assuming that the core height-to-diameter ratio α applies to the core dimensions including extrapolation lengths so that

$$\rho_o \cong \sqrt{\frac{D_c(2.47/\alpha^2 + 5.79}{\nu\Sigma_{f_c} - \Sigma_{a_c}}} - \delta_p \qquad (2.5.)$$

The second stage consists of solving Eq. (2.1.) by Newton's method, still assuming that the radial and axial reflector savings are equal to the plane geometry reflector savings obtained from Eq. (2.3.) and using the result of Eq. (2.5.) as the initial guess. The third stage in obtaining the critical radius is to solve Eq. (2.1.) by Newton's method for the case in which the axial and radial reflector savings are functions of the core radius (and the core height through the core height-to-diameter ratio α). For a cylindrical core with a radial reflector, the one-group criticality relation in the notation of Meghreblian and Holmes[(3)] is

$$\frac{D_c\lambda_c\, J_1\,(\lambda_c\,\rho_o)}{J_o\,(\lambda_c\,\rho_o)} - \frac{D_{rr}\,\lambda_r\, L_1\,(\rho_o)}{L_o\,(\rho_o)} = 0 \qquad (2.6.)$$

where

$$\lambda_c^2 \equiv B^2 - \left(\frac{\pi/2}{\alpha\rho_o + \delta_a}\right)^2 \qquad (2.7.)$$

$$\lambda_r^2 \equiv k_{rr}^2 + \left(\frac{\pi/2}{\alpha\rho_o + \delta_a}\right)^2 \qquad (2.8.)$$

$$k_{rr}^2 \equiv \Sigma a_{rr} / D_{rr} \qquad (2.9.)$$

$$L_1(\rho_o) \equiv I_o(\lambda_r\tilde{\rho}_1) K_1(\lambda_r\rho_o) + I_1(\lambda_r\rho_o) K_o(\lambda_r\tilde{\rho}_1) \qquad (2.10.)$$

$$L_o(\rho_o) \equiv I_o(\lambda_r\tilde{\rho}_1) K_o(\lambda_r\rho_o) - I_o(\lambda_r\rho_o) K_o(\lambda_r\tilde{\rho}_1) \qquad (2.11.)$$

$$\tilde{\rho}_1 = \rho_o + T_r + 2.13\ D_{rr} \quad , \qquad (2.12.)$$

and I_o, I_1, J_o, J_1, K_o, and K_1 represent various Bessel functions.

Note that for given values of the core radius ρ_o, the core height-to-diameter ratio α, the radial reflector thickness T_r, and the material properties, D_c, D_{rr}, B, and k_{rr}, that the set of equations (2.6.) through (2.12.) may be solved for the axial reflector savings δ_a. This solution is obtained by Newton's method in a separate subroutine.

For a cylindrical core with an end reflector, the one-group criticality relation is

$$D_c \mu_c \tan (\mu_c \alpha \rho_o) - D_{ar} \mu_r \coth (\mu_r \tilde{T}_a) \tag{2.13.}$$

$$\mu_c^2 \equiv B^2 - \left(\frac{2.405}{\rho_o + \delta_r}\right)^2 \tag{2.14.}$$

$$\mu_r^2 \equiv k_{ar}^2 + \left(\frac{2.405}{\rho_o + \delta_r}\right)^2 \tag{2.15.}$$

$$k_{ar}^2 \equiv \Sigma a_{ar} / D_{ar} \tag{2.16.}$$

$$\tilde{T}_a \equiv T_a + 2.13 D_{ar} \quad . \tag{2.17.}$$

As before, for given values of the core radius ρ_o, the core height-to-diameter ratio α, the axial reflector thickness T_a, and the material properties, the set of equations (2.13.) through (2.17.) may be solved for the radial reflector savings δ_r. This solution is obtained by Newton's method in a separate subroutine.

A special subroutine is used to solve equations (2.6.) through (2.12.) for the critical radius in cases in which the core height-to-diameter ratio is greater than three. In these cases the axial reflector savings required in equations (2.7.) and (2.9.) can be assumed to be the same as the reflector savings obtained from the plane geometry equation (2.3.) with negligible error.

3. TEST RESULTS

Tables (3.1.) and (3.2.) compare critical diameters for systems containing 90UC-10ZrC fuel. The values obtained from the solution of Eq. (2.1.) are designated as "new" values and the "old" values are from an older subroutine based upon interpolation in tables (one for each fissile nuclide) of diameters of infinite cylinders versus fissile nuclide density followed by a correction based upon the desired core height-to-diameter ratio. The one-group cross sections used in the new subroutine were averaged over the neutron spectrum used in the preliminary calculations for the Thermionic Critical Assembly. The fuel densities were chosen at the tabular points used in the original subroutine, to remove any uncertainties due to interpolation.

Table 3.1.

Comparison of Critical Diameter Estimates

For $90U^{233}$ C-10ZrC Systems

$V_f = 0.2964$

RHØF (g/cc)	D (old) (cm)	D (new) (cm)	$\frac{D(new) - D(old)}{D(old)} \times 100$
2.527	72.3	78.3	8.3
3.270	52.8	57.6	9.1
3.901	43.4	47.7	9.9
5.945	31.6	31.8	0.6
9.475	22.3	21.5	-3.6
18.579	13.4	13.3	-0.7

$V_f = 0.0759$

RhØF (g/cc)	D (old) (cm)	D (new) (cm)	$\frac{D(new) - D(old)}{D(old)} \times 100$
9.860	72.3	81.1	12.2
12.761	52.8	58.6	11.0
15.226	43.4	48.2	11.0

Table 3.2.

Comparison of Critical Diameter Estimates

for 90 $(93U^{235} - 7U^{238})$ C-10ZrC Systems

$V_f = 0.2962$

RHØF (g/cc)	D (old) (cm)	D (new) (cm)	$\frac{D\ (new = D\ (old)}{D\ (old)} \times 100$
4.511	82.3	79.5	-3.4
5.110	65.2	68.0	4.3
6.388	48.4	52.7	8.9
8.384	42.1	39.9	-5.2
10.180	39.2	33.2	-15.3

$V_f = 0.1266$

RHØF (g/cc)	D (old) (cm)	D (new) (cm)	$\frac{D\ (new) - D\ (old)}{D\ (old)} \times 100$
10.555	82.3	79.8	-3.0
11.956	65.2	68.1	4.4
14.945	48.4	52.6	8.7
19.615	42.1	39.8	-5.5

The agreement between the "old" and the "new" values seems satisfactory for survey purposes when one recalls that the "old" values are a composite of calculations for systems with varying reflector thicknesses, fuel densities and volume fractions, core height-to-diameter ratios, etc.

Table 3.3. gives a comparison of critical diameters calculated from Eq. (2.1.) with some of the critical diameters calculated with the GAPLSN transport theory code[5] in infinite cylindrical geometry at the time of the preparation of the original subroutine. The agreement between values in Table 3.3. is sometimes better and sometimes worse than in Tables 3.1. and 3.2.

Table 3.4. compares predicted critical diameters for a 90 ($93U^{235}$ - $7U^{238}$) C-10ZrC system as a function of core height-to-diameter ratio α. The older subroutine used the equation

$$\rho_o(\alpha) = \rho_o(\infty)\sqrt{1 + 0.427/\alpha^2} \tag{3.1.}$$

to estimate the critical radius for a finite cylinder from the interpolated value, $\rho_o(\infty)$, of the critical diameter of an infinite cylinder. The results shown in Table 3.4. indicate that this is not a very good approximation since the new prediction agrees well with the GAPLSN result (for $\alpha = \infty$) at large values of α but diverges for smaller values of α and there is good reason to believe that the new prediction subroutine incorporates a much more accurate representation of the geometrical variables for a finite cylinder.

Table 3.3.

Comparison of Critical Diameters With GAPLSN Results

Fuel Material	D (GAPLSN) (cm)	D (new) (cm)	$\frac{D\ (new) - D\ (GAPLSN)}{D\ (GAPLSN)} \times 100\%$
90 $U^{233}C$ - 10ZrC	18.84*	21.17	+12.5
90 $U^{235}C$ - 10ZrC	35.16	36.00	+ 2.4%
$U^{233}O_2$	22.65	25.33	+11.83
$U^{235}O_2$	44.14	44.57	+ 1.0
33 $U^{233}O_2$ - 67W	41.88	44.61	+ 6.5
33 $U^{235}O_2$ - 67W	~130.**	108.05	-16.9

* This point is very strange. The older subroutine prediction for this point was a diameter of about 22.7 cm which leads to a difference of -6.7% instead of +12.5%. The 18.84 cm value quoted here is the actual value from a GAPLSN problem with the proper atom densities, etc. that was run late in 1965.

** This is an estimate based on several unconverged GAPLSN calculations. With this fuel material (fuel volume fraction = 0.2370) the system is subcritical for core height-to-diameter ratios of two or less.

Table 3.4.

Effect of Core Height-to-Diameter Ratio

Core Height-to-Diameter Ratio	D (old) (cm)	D (new) (cm)	$\frac{D\ (new) - D\ (old)}{D\ (old)} \times 100$
2	36.99	40.49	+ 9.5
3	35.98	38.36	+ 6.6
4	35.63	37.59	+ 5.5
5	35.46	36.98	+ 4.3
10	35.23	36.00	+ 2.2

The subroutine has been used for over a year for parameter surveys with several different sets of one-group cross sections. The most convenient method of normalizing the results to specific design points is to divide $\nu\Sigma_f$ by an input value of k_{eff}. The values of k_{eff} required to force agreement with multigroup transport theory results have generally been in the range 1.0 to 1.15.

```
@P     HDG   TEST PROGRAM FOR CRITOM SUBROUTINE
@IWX   FOR   TEST/S1,TEST/S1,TEST/R1
C    TEST PROGRAM FOR CRITOM SUBROUTINE
C
        DIMENSION A(11)
        A(1)  =    .01259990
        A(2)  =    .02153081
        A(3)  =   1.0122988
        A(4)  =   1.0625
        A(5)  =    .06031171
        A(6)  =    .64561948
        A(7)  =   7.62
        A(8)  =    .013636
        A(9)  =    .75895464
        A(10) = 10.16
        A(11) =  2.
        CALL CRITOM( A, DC )
        HD = A(4)
        HC = HD * DC
        PRINT 1, DC, HC
      1 FORMAT(//6H DC = F12.6,1H,5X5H HC = F12.6)
        END
                                                                    % 0
@P     HDG   CRITOM/S1 CRITICALITY SUBROUTINE
@IWX   FOR   CRITOM/S1,CRITOM/S1,CRITOM/R1
C    SUBROUTINE TO ESTIMATE THE CRITICAL DIAMETER OF A CYLINDRICAL
C    REACTOR ACCORDING TO A ONE-GROUP, TWO-DIMENSIONAL, DIFFUSION THEORY
C    MODEL GIVEN THE FOLLOWING INPUT DATA
C     1. SIGAC  = CORE ABSORPTION CROSS SECTION (A.C.S.)
C     2. NUSIGF = NU TIMES THE FISSION CROSS SECTION FOR THE CORE
C     3. DCC    = CORE DIFFUSION COEFFICIENT (D.C.)
C     4. HD     = CORE HEIGHT-TO-DIAMETER RATIO
C     5. SIGARA = AXIAL REFLECTOR A.C.S.
C     6. DCRA   = AXIAL REFLECTOR D.C.
C     7. TRA    = AXIAL REFLECTOR THICKNESS
C     8. SIGARR = RADIAL REFLECTOR A.C.S.
C     9. DCRR   = RADIAL REFLECTOR D.C.
C    10. TRR    = RADIAL REFLECTOR THICKNESS
C    11. IP     = PRINT CONTROL VARIABLE
C                 0 = NO PRINTING
C                 2 = PRINT CROSS SECTIONS AND RESULTS
C                 3 = ABOVE PLUS DEBUG PRINT STATEMENTS
C
        SUBROUTINE CRITOM( A, DC )
        REAL KR, KRA, KRR, NUSIGF
        DIMENSION A(11)
        COMMON/COM1/BC,HD1,DH1,DR1,KRA,KRR,TRA1,TRR1,DCRA,DCRR,DCC,DELTA,
       1 IPP
C
C    COM1 TRANSMITS DATA TO DH, DHF, AND DR
C
C    DEFINE STATEMENT FUNCTIONS
C
        F1(R)=1./(HD*R+DELTA)
        F2(R)=1./(R+DELTA)
        F(R)=BC**2-2.4674*F1(R)**2-5.7831*F2(R)**2
```

```
      FP(R)=2.*(HD*2.4674*F1(R)**3+5.7831*F2(R)**3)
      DDH(DV3) = ( -DH(0.999*DV3) + DH(1.001*DV3) ) / (0.002*DV3)
      DDR(DV4) = ( -DR(0.999*DV4) + DR(1.001*DV4) ) / (0.002*DV4)
C
C   BEGIN SUBROUTINE
C
      SIGAC  = A(1)
      NUSIGF = A(2)
      DCC    = A(3)
      HD     = A(4)
      SIGARA = A(5)
      DCRA   = A(6)
      TRA    = A(7)
      SIGARR = A(8)
      DCRR   = A(9)
      TRR    = A(10)
      IP     = A(11)
C
      IF(IP.LT.1)GO TO 901
      PRINT 1000
 1000 FORMAT(/24H ENTER SUBROUTINE CRITOM  /)
 901  CONTINUE
C
      HD1 = HD
      TRA1 = TRA
      TRR1 = TRR
      TR = 0.5 * ( TRA1 + TRR1 )
      SIGAR  = (SIGARA + SIGARR) * 0.5
      DCR    = 2. / ( (1./DCRA) + (1./DCRR) )
      IPP = IP
C
C   PRINT AVERAGE CROSS SECTIONS
C
      IF(IP.LT.1)GO TO 904
      PRINT 1027,NUSIGF,SIGAC,DCC                                      9/14/67
 1027 FORMAT( 8H NUSIGC=1PE12.6,1H,6X8H SIGAC =E12.6,1H,6X8H DCC   =E12. 9/14/67
     16/)                                                              9/14/67
      PRINT 1013,SIGAR,DCR                                             9/14/67
 1013 FORMAT( 8H SIGAR =1PE12.6,1H,6X8H DCR   =E12.6/)                 9/14/67
      PRINT 1014, SIGARA, DCRA, SIGARR, DCRR
 1014 FORMAT( 8H SIGARA=1PE12.6,1H,6X8H DCRA  =E12.6 //   8H SIGARR=E12.
     16,1H,6X8H DCRR  =E12.6/)
      PRINT 1016, HD, TRA, TRR
 1016 FORMAT( 8H HD    =F12.6,1H,6X8H TRA   =F12.6,1H,6X8H TRR   =F12.6/
     1)
 904  CONTINUE
C
C   *************************************************************************
C
C   BEGIN CRITICAL SIZE CALCULATION
C
C    SLAB GEOMETRY REFLECTOR SAVINGS
C
      IF(SIGAC.GE.NUSIGF)GO TO 110                                     6/19/67
      BC=SQRT((NUSIGF-SIGAC)/DCC)
      KR=SQRT(SIGAR/DCR)
      KRA=SQRT(SIGARA/DCRA)
```

```
      KRR=SQRT(SIGARR/DCRR)
      DELTA=ATAN(TANH(KR*(TR+2.13*DCR))*DCC*BC/(DCR*KR) )/BC
C
C   PRINT BUCKLINGS AND SLAB GEOMETRY REFLECTOR SAVINGS
C
      IF(IP.LT.2)GO TO 905
      PRINT 1015,BC,KR,DELTA
 1015 FORMAT( 8H BC     =1PE12.6,1H,6X8H KR     =E12.6,1H,6X8H DELTA =E12. 9/14/67
     16/)                                                                   9/14/67
 905  CONTINUE
C
C   CRITICAL RADIUS USING SLAB GEOMETRY REFLECTOR SAVINGS FOR BOTH AXIAL
C   AND RADIAL DIRECTIONS
C
C   OBTAIN INITIAL ESTIMATE OF CORE RADIUS BY ASSUMING THAT THE CORE
C   HEIGHT-TO-DIAMETER RATIO APPLIES TO THE EXTRAPOLATED CORE DIMENSIONS
C   AND THEN USE NEWTON'S METHOD TO SOLVE THE CASE IN WHICH THE HEIGHT-
C   TO-DIAMETER RATIO APPLIES TO THE ACTUAL CORE DIMENSIONS
C
C   FUNCTIONS F(RC) AND FP(RC) ARE STATEMENT FUNCTIONS DEFINED AT THE
C   BEGINNING OF THE SUBROUTINE
C
      RC=SQRT(DCC*(2.47/HD**2+5.79)/(NUSIGF-SIGAC))-DELTA
      DO 70 K=1,100
      RC1=RC-F(RC)/FP(RC)
      IF(ABS((RC1-RC)/RC1).LT.1.E-6)GO TO 90
      IF(K.GT.98)GO TO 80
 70   RC=RC1
 80   PRINT 1021
 1021 FORMAT(//60H ***** NEWTONS METHOD FAILED TO FIND INITIAL CRITICAL
     1RADIUS / )
      RC=0.
      HC=0.
      RETURN
 90   RC=RC1
C
C   CRITICAL RADIUS USING CYLINDRICAL GEOMETRY REFLECTOR SAVINGS IN AN
C   ITERATIVE SCHEME
C
C   FUNCTION DH(RC) IS A FUNCTION SUBPROGRAM WHICH RETURNS A VALUE OF
C   THE AXIAL REFLECTOR SAVINGS FOR A GIVEN CORE RADIUS GIVEN THE
C   FOLLOWING ARGUMENTS(IN COMMON)
C       BC  = CORE MATERIAL BUCKLING( SQRT( (NSIGFC-SIGAC)/DCC ) )
C       HD  = CORE HEIGHT-TO-DIAMETER RATIO
C       DHI = INITIAL ESTIMATE OF AXIAL REFLECTOR SAVINGS
C       KR  = REFLECTOR MATERIAL CONSTANT( SQRT( SIGAR/DCR ) )
C       TRR = RADIAL REFLECTOR THICKNESS
C       DCR = REFLECTOR DIFFUSION CONSTANT
C       DCC = CORE DIFFUSION CONSTANT
C
C   FUNCTION DHF(RC) SOLVES THE SAME EQUATIONS AS FUNCTION DH(RC) FOR
C   THE CORE RADIUS INSTEAD OF THE AXIAL REFLECTOR SAVINGS IN CASES
C   WHEN THE CORE HEIGHT-TO-DIAMETER RATIO IS GREATER THAN 3.  THE AXIAL
C   REFLECTOR SAVINGS IS ASSUMED TO BE EQUAL TO THE SLAB GEOMETRY
C   REFLECTOR SAVINGS IN THIS CALCULATION.
C
C   FUNCTION DR(RC) IS A FUNCTION SUBPROGRAM WHICH RETURNS A VALUE OF
```

```
C   THE RADIAL REFLECTOR SAVINGS FOR A GIVEN CORE RADIUS GIVEN THE
C   SAME ARGUMENTS AS THE FUNCTION DH(RC) EXCEPT THAT NEED AN INITIAL
C   ESTIMATE FOR THE RADIAL REFLECTOR SAVINGS, DR1, INSTEAD OF DH1
C   AND THE AXIAL REFLECTOR THICKNESS, TRA, INSTEAD OF TRR
C
C
C   DEBUG PRINT STATEMENT FOLLOWS
C
      IF(IP.LT.3)GO TO 906
      PRINT 1018,RC
 1018 FORMAT(10H RC       =1PE14.7)
 906  CONTINUE
C
      RCS = RC
      DH1 = DELTA
      DR1 = DELTA
      C12 = 0.33*DELTA
      C13 = 2.72*DELTA
      M = 0
      RC = 0.97 * RC                                                    4/27/67
      IF( HD.LT.3.001 ) GO TO 95
         RC = DHF(RC)
         GO TO 210
 95   C1 = DH(RC)
 97   C2 = HD*RC + C1
      C3 = 1.5707963 / C2
      C4 = C3 * C3
      C5 = DR(RC)
      C6 = RC + C5
      C7 = 2.4048 / C6
      C8 = C7 * C7
      C9 = BC*BC - C4 - C8
 98   IF(C1.LT.C12.OR.C1.GT.C13.OR.C5.LT.C12.OR.C5.GT.C13)GO TO 101
         C10 = 2.*( C3*C4*(HD+DDH(RC)) + C7*C8*(1.+DDR(RC)) )
         GO TO 103
 101  IF(C5.LT.C12 .OR. C5.GT.C13)GO TO 102
         C10 = 4.*C7*C8*(1.+DDR(RC))
         GO TO 103
 102  IF(C1.LT.C12 .OR. C1.GT.C13)GO TO 110
         C10 = 4.*C3*C4*(HD+DDH(RC))
 103  C11 = C9/C10
      RC0 = RC - C11
C
C   NEWTON'S METHOD SOLUTION FOR THE CRITICAL RADIUS CAN OVERSHOOT FAR
C   ENOUGH ON THE FIRST ITERATION TO CAUSE NEGATIVE ARGUMENTS TO THE
C   SQUARE ROOT FUNCTIONS USED IN THE CALCULATION OF THE RADIAL AND
C   AXIAL REFLECTOR SAVINGS.  RMIN IS USED TO PREVENT THIS PROBLEM.
C
      C14=1.1*RC
      RC1=AMIN1(C14,RC0)
      C14=0.9*RC
      C15=1.0102*((1.5708/BC-C1)/HD+0.01)
      C16=1.0102*(2.405/BC-C5+0.01)
      RMIN=AMAX1(C14,C15,C16)
      RC1=AMAX1(RMIN,RC1)
C
C   DEBUG PRINT STATEMENTS FOLLOW
```

```
C
      IF(IP.LT.3)GO TO 907
      PRINT 104,C1,C2,C3,C4,C5,C6,C7,C8,C9,C10,RC0,RMIN
 104  FORMAT(/1P12E10.4)
      PRINT 105,RC1
 105  FORMAT(/10H RC      = 1PE14.7//////)
 907  CONTINUE
C
      IF(M.GT.99 .OR. RC1.LT..5*RCS .OR. RC1.GT.2.*RCS)GO TO 110
      IF(ABS((RC1-RC)/RC).LT.1.E-5.OR.ABS(C9).LT.1.E-6)GO TO 120
      RC = RC1
      M = M + 1
      GO TO 95
 110  PRINT 115
 115  FORMAT(//39H ***** COULD NOT FIND FINAL CORE RADIUS / )
      RC = 0.
      GO TO 210
 120  RC = RC1
 210  CONTINUE
C
C   CRITICAL DIAMETER AND HEIGHT
C
      DC=2.*RC
      HC=HD*DC
C
C   PRINT AXIAL AND RADIAL REFLECTOR SAVINGS PLUS CORE RADIUS, CORE
C   DIAMETER, AND CORE HEIGHT.
C
      IF(IP.LT.1)GO TO 908
      PRINT 1028,C1,C5,RC,DC,HC                                       9/14/67
 1028 FORMAT( 8H DH    =1PE12.6,1H,6X8H DR    =E12.6 //    8H RC    =E12. 9/14/67
     16,1H,6X8H DC    =E12.6,1H,6X8H HC    =E12.6/)                   9/14/67
 908  CONTINUE
C
      IF(IP.LT.1)GO TO 909
      PRINT 2000
 2000 FORMAT(  25H END OF SUBROUTINE CRITDM    )
 909  CONTINUE
C
      RETURN
      END


@P    HDG   DH/S1 RADIAL CALCULATION
@IWX  FOR   DH/S1,DH/S1,DH/R1
C   FUNCTION SUBPROGRAM TO CALCULATE THE AXIAL(END) REFLECTOR SAVINGS
C   FROM THE ONE-GROUP TWO-REGION DIFFUSION THEORY CRITICALITY CONDITION
C   FOR A CYLINDRICAL CORE WITH A RADIAL(SIDE) REFLECTOR
C
C   THE FUNCTIONS FJ0,FJ1,FK0,FK1,FI0, AND FI1 ARE FUNCTION SUBPROGRAMS
C   WHICH RETURN VALUES OF THE J0,J1,K0,K1,I0, AND I1 BESSEL FUNCTIONS
C

      FUNCTION DH(DRV)
      REAL KR, KRA
      COMMON/COM1/BC,HD,DH1,DR1,KRA,KR,TRA1,TR,DCRA,DCR,DCC,DELTA,IP
C
```

```
C     DEFINE STATEMENT FUNCTIONS
        FLC(DV1) = SQRT( BC*BC - (1.5707963/(HD*RC + DV1) )**2 )
        FLR(DV2) = SQRT( KR*KR + (1.5707963/(HD*RC + DV2) )**2 )
        R1(DV3) = FLR(DV3) * (RC + TR + 2.1312*DCR)
        R0(DV4) = FLR(DV4) * RC
        FL0(DV5) =  FI0(R1(DV5))*FK0(R0(DV5)) - FI0(R0(DV5))*FK0(R1(DV5))
        FL1(DV6) =  FI0(R1(DV6))*FK1(R0(DV6)) + FI1(R0(DV6))*FK0(R1(DV6))
        SIDEC(DV7) = DCC * FLC(DV7) * FJ1(FLC(DV7)*RC) / FJ0(FLC(DV7)*RC)
       1 - DCR * FLR(DV7) * FL1(DV7) / FL0(DV7)
        DERIVS(DV8) = ( -SIDEC(0.99*DV8-1.E-5) + SIDEC(1.01*DV8+1.E-5) )
       1 / ( 0.02*DV8 + 2.E-5 )
C
        RC = DRV
        C12=1.5708/BC-HD*(0.98*RC-2.E-5)+0.02
        DH1=AMAX1(C12,DH1)
        C13=0.3*DELTA
        C14=3.*DELTA
C
C     DEBUG PRINT STATEMENTS FOLLOW
C
        IF(IP.LT.3)GO TO 901
        PRINT 100,BC,HD,DH1,DR1,KR,TR,DCR,DCC,RC
  100   FORMAT(///1P9E10.4/)
        PRINT 10,DH1
  10    FORMAT( 10H DH      = 1PE14.7 )
  901   CONTINUE
C
C     SOLVE FOR AXIAL REFLECTOR SAVINGS USING NEWTON'S METHOD
C
        K = 1
  20    C1 = FLC(DH1)
        C2 = FJ1(C1*RC)
        C3 = FJ0(C1*RC)
        C4 = DCC*C1*C2/C3
        C5 = FLR(DH1)
        C6 = FL1(DH1)
        C7 = FL0(DH1)
        C8 = DCR*C5*C6/C7
        C9 = C4 - C8
        C10 = DERIVS(DH1)
        C11 = C9 / C10
        DH2 = DH1 - C11
        C15=1.4*DH1
        DH2=AMIN1(C14,C15,DH2)
        C15=0.7*DH1
        DH2=AMAX1(C12,C13,C15,DH2)
C
C     DEBUG PRINT STATEMENTS FOLLOW
C
        IF(IP.LT.3)GO TO 902
        PRINT 110,C1,C2,C3,C4,C5,C6,C7,C8,C9,C10,C11
  110   FORMAT(/1P11E10.4)
        PRINT 15,DH2
  15    FORMAT( 10H DH      = 1PE14.7 )
  902   CONTINUE
C
```

```
      IF(K.GT.25 .OR. DH2.LT.0.)GO TO 30
      IF(ABS((DH2-DH1)/DH1).LT.1.E-6.OR.ABS(C9).LT.1.E-7)GO TO 50
      DH1 = DH2
      K = K + 1
      GO TO 20
 30   PRINT 40
 40   FORMAT(//63H ***** NEWTONS METHOD FAILED FOR AXIAL(END) REFLECTOR
     1SAVINGS    / )
      DH = -1.E+5
      RETURN
 50   DH1 = DH2
      DH = DH1
      RETURN
      END


@P    HDG   DHF/S1 RADIAL CALCULATION FOR H/D .GT. 3.
@IWX  FOR   DHF/S1,DHF/S1,DHF/R1
C   FUNCTION SUBPROGRAM TO CALCULATE THE CRITICAL RADIUS FROM THE ONE-
C   GROUP TWO-REGION DIFFUSION THEORY CRITICALITY CONDITION FOR A
C   CYLINDRICAL CORE WITH A RADIAL(SIDE) REFLECTOR AND A SPECIFIED
C   CONSTANT VALUE OF THE AXIAL REFLECTOR SAVINGS
C
C   THE FUNCTIONS FJ0,FJ1,FK0,FK1,FI0, AND FI1 ARE FUNCTION SUBPROGRAMS
C   WHICH RETURN VALUES OF THE J0,J1,K0,K1,I0, AND I1 BESSEL FUNCTIONS
C
C
      FUNCTION DHF(DRV)
      REAL KR, KRA
      COMMON/COM1/BC,HD,DH1,DR1,KRA,KR,TRA1,TR,DCRA,DCR,DCC,DELTA,IP
C
C   DEFINE STATEMENT FUNCTIONS
C
      FLC(DV1) = SQRT( BC*BC - (1.5707963/(HD*DV1 + DELTA) )**2 )
      FLR(DV2) = SQRT( KR*KR + (1.5707963/(HD*DV2 + DELTA) )**2 )
      R1(DV3) = FLR(DV3) * (DV3 + TR + 2.1312*DCR)
      R0(DV4) = FLR(DV4) * DV4
      FL0(DV5) = FI0(R1(DV5))*FK0(R0(DV5)) - FI0(R0(DV5))*FK0(R1(DV5))
      FL1(DV6) = FI0(R1(DV6))*FK1(R0(DV6)) + FI1(R0(DV6))*FK0(R1(DV6))
      SIDEC(DV7) = DCC * FLC(DV7) * FJ1(FLC(DV7)*DV7) / FJ0(FLC(DV7)*DV7
     1)- DCR * FLR(DV7) * FL1(DV7) / FL0(DV7)
      DERIVS(DV8) = ( -SIDEC(0.99*DV8) + SIDEC(1.01*DV8))/ (0.02*DV8)
C
      RC = DRV
      C12 = 1.0102*((1.5708/BC-DELTA)/HD+0.01)
      RC = AMAX1(C12,RC)
C
C   DEBUG PRINT STATEMENT FOLLOWS
C
      IF(IP.LT.3)GO TO 901
      PRINT 100,BC,HD,DH1,DR1,KR,TR,DCR,DCC,DELTA,DRV,C12,RC
 100  FORMAT(///1P12E10.4/)
 901  CONTINUE
C
C   SOLVE FOR CORE RADIUS USING NEWTONS METHOD
C
      K = 1
```

```
 20     C1 = FLC(RC)
        C2 = FJ1(C1*RC)
        C3 = FJ0(C1*RC)
        C4 = DCC*C1*C2/C3
        C5 = FLR(RC)
        C6 = FL1(RC)
        C7 = FL0(RC)
        C8 = DCR*C5*C6/C7
        C9 = C4 - C8
        C10 = DERIVS(RC)
        C11 = C9/C10
        RC1 = RC - C11
        C15 = 1.1*RC
        RC1 = AMIN1(C15,RC1)
        C15 = 0.9*RC
        RC1 = AMAX1(C12,C15,RC1)
C
C   DEBUG PRINT STATEMENTS FOLLOW
C
        IF(IP.LT.3)GO TO 902
        PRINT 110,C1,C2,C3,C4,C5,C6,C7,C8,C9,C10,C11
 110    FORMAT(/1P11E10.4)
        PRINT 15,RC1
 15     FORMAT( 10H RC      = 1PE14.7 )
 902    CONTINUE
C
        IF(K.GT.50 .OR. RC1.LT..5*DRV .OR. RC1.GT.2.*DRV)GO TO 30
        IF(ABS((RC1-RC)/RC).LT.1.E-6 .OR. ABS(C9).LT.1.E-7)GO TO 50
        RC = RC1
        K = K + 1
        GO TO 20
 30     PRINT 40
 40     FORMAT(//49H ***** NEWTON'S METHOD FAILED FOR CRITICAL RADIUS   )
        DHF=0.
        RETURN
 50     RC = RC1
        DHF=RC
        RETURN
        END


@P      HDG    DR/S1 AXIAL CALCULATION
@IWX    FOR    DR/S1,DR/S1,DR/R1
C   FUNCTION SUBPROGRAM TO CALCULATE THE RADIAL(SIDE) REFLECTOR SAVINGS
C   FROM THE ONE-GROUP TWO-REGION DIFFUSION THEORY CRITICALITY CONDITION
C   FOR A CYLINDRICAL CORE WITH A AXIAL(END) REFLECTOR
C
C
        FUNCTION DR(DRV)
        REAL KR, KRR
        COMMON/COM1/BC,HD,DH1,DR1,KR,KRR,TR,TRR1,DCR,DCRR,DCC,DELTA,IP
C
C   DEFINE STATEMENT FUNCTIONS
C
        FMUC(DV1) = SQRT( BC*BC - ( 2.4048/(RC + DV1) )**2 )
        FMUR(DV2) = SQRT( KR*KR + ( 2.4048/(RC + DV2) )**2 )
        COTH(DV3) = 1./ TANH( FMUR(DV3) * (TR + 2.1312*DCR) )
```

```
      ENDC(DV4) = DCC * FMUC(DV4) * TAN( FMUC(DV4) * HD * RC )
     1 - DCR * FMUR(DV4) * COTH(DV4)
      DERIVE(DV5) = ( -ENDC(0.99*DV5-1.E-5) + ENDC(1.01*DV5+1.E-5) )
     1 / ( 0.02*DV5 + 2.E-5 )
C
      RC = DRV
      C10=2.405/BC-0.98*RC+0.02
      DR1=AMAX1(C10,DR1)
      C11=0.3*DELTA
      C12=3.*DELTA
C
C   DEBUG PRINT STATEMENTS FOLLOW
C
      IF(IP.LT.3)GO TO 901
      PRINT 100,BC,HD,DH1,DR1,KR,TR,DCR,DCC,RC
 100  FORMAT(///1P9E10.4/)
      PRINT 10,DR1
 10   FORMAT( 10H DR      = 1PE14.7 )
 901  CONTINUE
C
C   SOLVE FOR RADIAL REFLECTOR SAVINGS USING NEWTON'S METHOD
C
      K = 1
 20   C1 = FMUC(DR1)
      C2 = TAN(C1*HD*RC)
      C3 = DCC*C1*C2
      C4 = FMUR(DR1)
      C5 = COTH(DR1)
      C6 = DCR*C4*C5
      C7 = C3 - C6
      C8 = DERIVE(DR1)
      C9 = C7 / C8
      DR2 = DR1 - C9
      C13=1.4*DR1
      DR2=AMIN1(C12,C13,DR2)
      C13=0.7*DR1
      DR2=AMAX1(C10,C11,C13,DR2)
C
C   DEBUG PRINT STATEMENTS FOLLOW
C
      IF(IP.LT.3)GO TO 902
      PRINT 110,C1,C2,C3,C4,C5,C6,C7,C8,C9
 110  FORMAT(/1P9E10.4)
      PRINT 15,DR2
 15   FORMAT( 10H DR      = 1PE14.7 )
 902  CONTINUE
C
      IF(K.GT.25 .OR. DR2.LT.0.)GO TO 30
      IF(ABS((DR2-DR1)/DR1).LT.1.E-6.OR.ABS(C7).LT.1.E-7)GO TO 50
      DR1 = DR2
      K = K + 1
      GO TO 20
 30   PRINT 40
 40   FORMAT(//63H ***** NEWTONS METHOD FAILED FOR RADIAL(SIDE) REFLECTO
     1R SAVINGS / )
      DR = -1.E+5
      RETURN
```

```
50    DR1 = DR2
      DR = DR1
      RETURN
      END
```

```
@P     HDG    FI0/S1 I0 BESSEL FUNCTION FOR X.GT.-3.75
@IWX   FOR    FI0/S1,FI0/S1,FI0/R1
C   FUNCTION SUBPROGRAM TO CALCULATE THE I0 BESSEL FUNCTION FOR X.GT.-3.75
C   REFERENCE - N.B.S. H.B.(1964), P.378
      FUNCTION FI0(X)
      IF(X.GT.-3.75)GO TO 10
      CALL EXIT
 10   IF(X.GT.3.75)GO TO 20
      T=(X/3.75)**2
      FI0=1.+T*(3.5156229+T*(3.0899424+T*(1.2067492+T*(.2659732+T*(
     1    .0360768+T*.0045813)))))
      RETURN
   20 T=3.75/X
      FI0=SQRT(1./X)*EXP(X)*(.39894228+T*(.01328592+T*(.00225319+T*(
     1    -.00157565+T*(.00916281+T*(-.02057706+T*(.02635537+T*(
     2    -.01647633+T*.00392377))))))))
      RETURN
      END
```

```
@P     HDG    FI1/S1 I1 BESSEL FUNCTION FOR X.GT.-3.75
@IWX   FOR    FI1/S1,FI1/S1,FI1/R1
C  FUNCTION SUBPROGRAM TO CALCULATE THE I1 BESSEL FUNCTION FOR X.GT.-3.75
C  REFERENCE - N.B.S. H.B.(1964), P.378
      FUNCTION FI1(X)
      IF(X.GT.-3.75)GO TO 10
      CALL EXIT
 10   IF(X.GT.3.75)GO TO 20
      T=(X/3.75)**2
      FI1=X*(.5+T*(.87890594+T*(.51498869+T*(.15084934+T*(.02658733+T*(
     1    .00301532+T*.00032411))))))
      RETURN
   20 T=3.75/X
      FI1=SQRT(1./X)*EXP(X)*(.39894228+T*(-.03988024+T*(-.00362018+T*(
     1    .00163801+T*(-.01031555+T*(.02282967+T*(-.02895312+T*(
     2    .01787654-T*.00420059))))))))
      RETURN
      END
```

```
@P     HDG    FJ0/S1 J0 BESSEL FUNCTION FOR X.GT.-3.
@IWX   FOR    FJ0/S1,FJ0/S1,FJ0/R1
C   FUNCTION SUBPROGRAM TO CALCULATE THE J0 BESSEL FUNCTION FOR X.GT.-3.
C   REFERENCE - N.B.S. H.B.(1964), PP.369-370
      FUNCTION FJ0(X)
      IF(X.GT.-3.)GO TO 10
      CALL EXIT
 10   IF(X.GT.3.)GO TO 20
      T=(X/3.)**2
      FJ0=1.+T*(-2.2499997+T*(1.2656208+T*(-.3163866+T*(.0444479+T*(
     1    -.0039444+T*.0002100)))))
```

```
      RETURN
   20 T=3./X
      F0=.79788456+T*(-.00000077+T*(-.00552740+T*(-.00009512+T*(
     1   .00137237+T*(-.00072805+T*.00014476)))))
      THETA0=X-.78539816+T*(-.04166397+T*(-.00003954+T*(.00262573+T*(
     1        -.00054125+T*(-.00029333+T*.00013558)))))
      FJ0=SQRT(1./X)*F0*COS(THETA0)
      RETURN
      END


@P    HDG   FJ1/S1 J1 BESSEL FUNCTION FOR X.GT.-3.
@IWX  FOR   FJ1/S1,FJ1/S1,FJ1/R1
C   FUNCTION SUBPROGRAM TO CALCULATE THE J1 BESSEL FUNCTION FOR X.GT.-3.
C   REFERENCE - N.B.S. H.B.(1964), P.370
      FUNCTION FJ1(X)
      IF(X.GT.-3.)GO TO 10
      CALL EXIT
 10   IF(X.GT.3.)GO TO 20
      T=(X/3.)**2
      FJ1=X*(.5+T*(-.56249985+T*(.21093573+T*(-.03954289+T*(.00443319+T*
     1    (-.00031761+T*.00001109))))))
      RETURN
   20 T=3./X
      F1=.79788456+T*(.00000156+T*(.01659667+T*(.00017105+T*(-.00249511
     1   +T*(.00113653-T*.00020033)))))
      THETA1=X-2.35619449+T*(.12499612+T*(.00005650+T*(-.00637879+T*(
     1        .00074348+T*(.00079824-T*.00029166)))))
      FJ1=SQRT(1./X)*F1*COS(THETA1)
      RETURN
      END


@P    HDG   FK0/S1 K0 BESSEL FUNCTION FOR X.GT.0.
@IWX  FOR   FK0/S1,FK0/S1,FK0/R1
C   FUNCTION SUBPROGRAM TO CALCULATE THE K0 BESSEL FUNCTION FOR X.GT.0.
C   REFERENCE - N.B.S. H.B.(1964), P.379
      FUNCTION FK0(X)
      IF(X.GT.0.)GO TO 10
      CALL EXIT
 10   IF(X.GT.2.)GO TO 20
      T=(X/2.)**2
      FK0=-ALOG(X/2.)*FI0(X)-.57721566+T*(.42278420+T*(.23069756+T*(
     1    .03488590+T*(.00262698+T*(.00010750+T*.00000740)))))
      RETURN
   20 T=2./X
      FK0=SQRT(1./X)*EXP(-X)*(1.25331414+T*(-.07832358+T*(.02189568+T*(
     1    -.01062446+T*(.00587872+T*(-.00251540+T*.00053208))))))
      RETURN
      END


@P    HDG   FK1/S1 K1 BESSEL FUNCTION FOR X.GT.0.
@IWX  FOR   FK1/S1,FK1/S1,FK1/R1
C   FUNCTION SUBPROGRAM TO CALCULATE THE K1 BESSEL FUNCTION FOR X.GT.0.
C   REFERENCE - N.B.S. H.B.(1964), P.379
      FUNCTION FK1(X)
```

```
      IF(X.GT.0.)GO TO 10
      CALL EXIT
 10   IF(X.GT.2.)GO TO 20
      T=(X/2.)**2
      FK1=ALOG(X/2.)*FI1(X)+(1./X)*(1.+T*(.15443144+T*(-.67278579+T*(
     1     -.18156897+T*(-.01919402+T*(-.00110404-T*.00004686))))))
      RETURN
   20 T=2./X
      FK1=SQRT(1./X)*EXP(-X)*(1.2533141+T*(.23498619+T*(-.03655620+T*(
     1     .01504268+T*(-.00780353+T*(.00325614-T*.00068245))))))
      RETURN
      END

@P    HDG   SAMPLE PROBLEM FOR CRITDM SUBROUTINE
@     XQT TEST
@BELR PMD
```

```
ENTER SUBROUTINE CRITDM

NUSIGC= 2.133081-02,        SIGAC = 1.259496-02,        DCC    = 1.612291+00

SIGAR = 6.873855-03,        DCR    = 7.101264-01

SIGARA= 3.117100-04,        DCRA   = 6.456195-01

SIGARR= 1.343600-02,        DCRR   = 7.889546-01

HD     =    1.322500,       TRA    =    7.620000,       TRR    =   10.160000

BC     = 7.360900-02,       KR     = 9.838578-02,       DELTA = 1.248283+01

DH     = 1.772720+01,       DR     = 1.237093+01

RC     = 2.386759+01,       DC     = 4.773519+01,       HC     = 6.312979+01

END OF SUBROUTINE CRITDM

DC =      47.735188,       HC =     63.129786


STOP CALLED FROM 014045
```

REFERENCES

1. Pidd, R. W., et al, "Thermionic Project Annual Summary Report," GA-7499 (1966) (C/RD).

2. Abramowitz, M. and Stegun, I. A., "Handbook of Mathematical Functions", Dover Publ., Inc., New York (1965), pp. 369-370 and pp. 378-379.

3. Meghreblian, R. V. and Holmes, D. K., "Reactor Analysis", Mc-Graw-Hill Book Co., Inc., New York (1960), pp. 418-479.

4. Shanstrom, R., et al, "Conceptual Design of a Thermionic Critical Equipment for Doppler Coefficient Measurements", GA-7472 (1966).

5. Alexander, J. H., et al, "GAPLSN, A Modified DSN Program for the Solution of the One-Dimensional Anisotropic Transport Equation," GA-4972 (1964).